
Figure 9.3. Environment after evaluating (define inc (make-adder 1)).

Figure 9.4. Environment for evaluating the body of (inc 149).

shown in Figure 9.3. The name inc has a value that is the procedure resulting
from the application of (make-adder 1). To evaluate the application, we follow
the application rule above and create a new environment containing a frame
with the parameter name, inc, and its associated operand value, 1.

The result of the application is the value of evaluating its body in this new en-
vironment. Since the body is a lambda expression, it evaluates to a procedure.
That procedure was created in the execution environment that was created to
evaluate the application of make-adder , hence, its environment pointer points
to the application environment which contains a place named inc holding the
value 1.

Next, consider evaluating (inc 149). Figure 9.4 illustrates the environment for
evaluating the body of the inc procedure. The evaluation creates a new envi-
ronment with a frame containing the place n and its associated value 149. We
evaluate the body of the procedure, (+ n v), in that environment. The value of
n is found in the execution environment. The value of v is not found there, so
evaluation continues by looking in the parent environment. It contains a place
v containing the value 1.

Exercise 9.1. Devise a Scheme expression that could have four possible values,
depending on the order in which application subexpressions are evaluated.



186 9.3. Mutable Pairs and Lists

Exercise 9.2. Draw the environment that results after evaluating:

> (define alpha 0)
> (define beta 1)
> (define update-beta! (lambda () (set! beta (+ alpha 1)))
> (set! alpha 3)
> (update-beta!)
> (set! alpha 4)

Exercise 9.3. Draw the environment that results after evaluating the following
expressions, and explain what the value of the final expression is. (Hint: first,
rewrite the let expression as an application.)

> (define (make-applier proc) (lambda (x) (proc x))
> (define p (make-applier (lambda (x) (let ((x 2)) x))))
> (p 4)

9.3 Mutable Pairs and Lists
The Pair datatype introduced in Chapter 5 is immutable. This means that onceimmutable

a Pair is created, the values in its cells cannot be changed.2

The MutablePair datatype is a mutable pair. A MutablePair is constructed using
mcons, which is similar to cons but produces a MutablePair. The parts of a Mu-
tablePair can be extracted using the mcar and mcdr procedures, which behave
analogously to the car and cdr procedures. A MutablePair is a distinct datatype
from a Pair; it is an error to apply car to a MutablePair, or to apply mcar to an
immutable Pair.

The MutablePair datatype also provides two procedures that change the values
in the cells of a MutablePair:

set-mcar! : MutablePair× Value→ Void
Replaces the value in the first cell of the MutablePair with the value of the
second input.

set-mcdr! : MutablePair× Value→ Void
Replaces the value in the second cell of the MutablePair with the value of
the second input.

The Void result type indicates that set-mcar! and set-mcdr! produce no output.

Here are some interactions using a MutablePair:

> (define pair (mcons 1 2))
> (set-mcar! pair 3)
> pair
(3 . 2)
2The mutability of standard Pairs is quite a controversial issue. In most Scheme implementations

and the standard definition of Scheme, a standard cons pair is mutable. But, as we will see later in
the section, mutable pairs cause lots of problems. So, the designers of DrRacket decided for Version
4.0 to make the standard Pair datatype immutable and to provide a MutablePair datatype for use
when mutation is needed.

> (set-mcdr! pair 4)
> pair
(3 . 4)

The set-mcdr! procedure allows us to create a pair where the second cell of the
pair is itself: (set-mcdr! pair pair). This produces the rather frightening object
shown in Figure 9.5. Every time we apply mcdr to pair , we get the same pair as

Figure 9.5. Mutable pair created by evaluating (set-mcdr! pair pair).

the output. Hence, the value of (mcar (mcdr (mcdr (mcdr pair)))) is 3.

We can also create objects that combine mutable and immutable Pairs. For ex-
ample, (define mstruct (cons (mcons 1 2) 3)) defines mstruct as an immutable
Pair containing a MutablePair in its first cell. Since the outer Pair is immutable,
we cannot change the objects in its cells. Thus, the second cell of mstruct always
contains the value 3. We can, however, change the values in the cells of the mu-
table pair in its first cell. For example, (set-mcar! (car mstruct) 7) replaces the
value in the first cell of the MutablePair in the first cell of mstruct .

Mutable Lists. As we used immutable Pairs to build immutable Lists, we can
use MutablePairs to construct MutableLists. A MutableList is either null or a Mu-
tablePair whose second cell contains a MutableList.

The MutableList type is defined by a library. To use it, evaluate the following
expression: (require racket/mpair). All of the examples in this chapter assume
this expression has been evaluated. This library defines the mlist procedure
that is similar to the list procedure, but produces a MutableList instead of an
immutable List. For example, (mlist 1 2 3) produces the structure shown in Fig-
ure 9.6. Each node in the list is a MutablePair, so we can use the set-mcar! and

Figure 9.6. MutableList created by evaluating (mlist 1 2 3).

set-mcdr! procedures to change the values in the cells.

> (define m1 (mlist 1 2 3))
> (set-mcar! (mcdr m1) 5)
> (set-mcar! (mcdr (mcdr m1)) 0)
> m1
{1 5 0} ; DrRacket denotes MutableLists using curly brackets.

Many of the list procedures from Chapter 5 can be directly translated to work on
mutable lists. For example, we can define mlist-length as:

(define (mlist-length m)
(if (null? m) 0 (+ 1 (mlist-length (mcdr m)))))

As shown in Exercise 9.4, though, we need to be careful when using mcdr to
recurse through a MutableList since structures created with MutablePairs can
include circular pointers.

Exercise 9.4. What is the value of (mlist-length pair) for the pair shown in Fig-
ure 9.5?

Exercise 9.5. [?] Define a mpair-circular? procedure that takes a MutablePair as
its input and outputs true when the input contains a cycle and false otherwise.

9.4 Imperative Programming
Mutation enables a style of programming known as imperative programming .imperative

programming Whereas functional programming is concerned with defining procedures that
can be composed to solve a problem, imperative programming is primarily con-
cerned with modifying state in ways that lead to a state that provides a solution
to a problem.

The main operation in function programming is application. A functional pro-
gram applies a series of procedures, passing the outputs of one application as
the inputs to the next procedure application. With imperative programming, the
primary operation is assignment (performed by set!, set-mcar! , and set-mcdr! in
Scheme; but typically by an assignment operator, often := or =, in languages de-
signed for imperative programming such as Pascal, Algol60, Java, and Python).

The next subsection presents imperative-style versions of some of the proce-
dures we have seen in previous chapters for manipulating lists. The following
subsection introduces some imperative control structures.

9.4.1 List Mutators
All the procedures for changing the value of a list in Section 5.4.3 actually do
not change any values; instead they construct new lists. When our goal is only
to change some elements in an existing list, this wastes memory constructing a
new list and may require more running time than a procedure that modifies the
input list instead. Here, we revisit some of the procedures from Section 5.4.3,
but instead of producing new lists with the desired property these procedures
modify the input list.

Example 9.2: Mapping

The list-map procedure (from Example 5.4) produces a new list that is the result
of applying the same procedure to every element in the input list.

(define (list-map f p)
(if (null? p) null (cons (f (car p)) (list-map f (cdr p)))))

Whereas the functional list-map procedure uses cons to build up the output list,
the imperative mlist-map! procedure uses set-car! to mutate the input list’s ele-
ments:

(define (mlist-map! f p)
(if (null? p) (void)

(begin (set-mcar! p (f (mcar p)))
(mlist-map! f (mcdr p)))))

The base case uses (void) to evaluate to no value. Unlike list-map which evalu-
ates to a List, mlist-map! is evaluated for its side effects and produces no output.

Assuming the procedure passed as f has constant running time, the running
time of the mlist-map! procedure is in Θ(n) where n is the number of elements
in the input list. There will be n recursive applications of mlist-map! since each
one passes in a list one element shorter than the input list, and each application
requires constant time. This is asymptotically the same as the list-map proce-
dure, but we would expect the actual running time to be faster since there is no
need to construct a new list.

The memory consumed is asymptotically different. The list-map procedure al-
locates n new cons cells, so it requires memory in Θ(n) where n is the number of
elements in the input list. The mlist-map! procedure is tail recursive (so no stack
needs to be maintained) and does not allocate any new cons cells, so it requires
constant memory.

Example 9.3: Filtering

The list-filter procedure takes as inputs a test procedure and a list and outputs
a list containing the elements of the input list for which applying the test proce-
dure evaluates to a true value. In Example 5.5, we defined list-filter as:

(define (list-filter test p)
(if (null? p) null

(if (test (car p)) (cons (car p) (list-filter test (cdr p)))
(list-filter test (cdr p)))))

An imperative version of list-filter removes the unsatisfying elements from a
mutable list. We define mlist-filter! using set-mcdr! to skip over elements that
should not be included in the filtered list:

(define (mlist-filter! test p)
(if (null? p) null

(begin (set-mcdr! p (mlist-filter! test (mcdr p)))
(if (test (mcar p)) p (mcdr p)))))

Assuming the test procedure has constant running time, the running time of the
mlist-filter! procedure is linear in the length of the input list. As with mlist-map! ,
the space used by mlist-filter! is constant, which is better than the Θ(n) space
used by list-filter .

Unlike mlist-map! , mlist-filter! outputs a value. This is needed when the first
element is not in the list. Consider this example:

> (define a (mlist 1 2 3 1 4))
> (mlist-filter! (lambda (x) (> x 1)) a)
{2 3 4}
> a
{1 2 3 4}

The value of a still includes the initial 1. There is no way for the mlist-filter! pro-



190 9.4. Imperative Programming

cedure to remove the first element of the list: the set-mcar! and set-mcdr! proce-
dures only enable us to change what the mutable pair’s components contain.

To avoid this, mlist-filter! should be used with set! to assign the variable to the
resulting mutable list:

(set! a (mlist-filter! (lambda (x) (> x 1)) a))

Example 9.4: Append

The list-append procedure takes as input two lists and produces a list consisting
of the elements of the first list followed by the elements of the second list. An
imperative version of this procedure instead mutates the first list to contain the
elements of both lists.

(define (mlist-append! p q)
(if (null? p) (error "Cannot append to an empty list")

(if (null? (mcdr p)) (set-mcdr! p q)
(mlist-append! (mcdr p) q))))

The mlist-append! procedure produces an error when the first input is null —
this is necessary since if the input is null there is no pair to modify.3

Like list-append, the running time of the mlist-append! procedure is in Θ(n)
where n is the number of elements in the first input list. The list-append pro-
cedure copies the first input list, so its memory use is in Θ(n) where n is the
number of elements in the first input list. The memory use of mlist-append! is
constant: it does not create any new cons cells to append the lists.

Aliasing. Adding mutation makes it possible to define many procedures more
efficiently and compactly, but introduces many new potential pitfalls in produc-
ing reliable programs. Since our evaluation model now depends on the environ-
ment in which an expression is evaluated, it becomes much harder to reason
about code by itself.

One challenge introduced by mutation is aliasing . There may be different waysaliasing

to refer to the same object. This was true before mutation also, but didn’t mat-
ter since the value of an object never changed. Once object values can change,
however, aliasing can lead to surprising behaviors.

For example,

> (define m1 (mlist 1 2 3))
> (define m2 (mlist 4 5 6))
> (mlist-append! m1 m2)
> (set! m1 (mlist-filter! (lambda (el) (= (modulo el 2) 0)) m1))

The value of m2 was defined as {4 5 6}, and no expressions since then explicitly
modified m2. But, the value of m2 has still changed! It changed because after
evaluating (mlist-append! m1 m2) the m1 object shares cells with m2. Thus,

3The mappend! library procedure in DrRacket takes a different approach: when the first input
is null it produces the value of the second list as output in this case. This has unexpected behavior
when an expression like (append! a b) is evaluated where the value of a is null since the value of a is
not modified.

when the mlist-filter! application changes the value of m1, it also changes the
value of m2 to {4 6}.

The built-in procedure eq? takes as input any two objects and outputs a Boolean.
The result is true if and only if the inputs are the same object. For example, (eq?
3 3) evaluates to true but (eq? (mcons 1 2) (mcons 1 2)) evaluates to false. Even
though the input pairs have the same value, they are different objects—mutating
one of the pairs does not effect the value of the other pair.

For the earlier mlist-append! example, (eq? m1 m2) evaluates to false since m1
and m2 do not refer to the same object. But, (eq? (mcdr m1) m2) evaluates to
true since the second cell of m1 points to the same object as m2. Evaluating
(set-mcar! m2 3) changes the value of both m1 and m2 since the modified cell is
common to both structures.

Exercise 9.6. Define an imperative-style procedure, mlist-inc! that takes as in-
put a MutableList of Numbers and modifies the list by adding one to the value
of each element in the list.

Exercise 9.7. [?] Define a procedure mlist-truncate! that takes as input a Mu-
tableList and modifies the list by removing the last element in the list. Specify
carefully the requirements for the input list to your procedure.

Exercise 9.8. [?] Define a procedure mlist-make-circular! that takes as input a
MutableList and modifies the list to be a circular list containing all the elements
in the original list. For example, (mlist-make-circular! (mlist 3)) should produce
the same structure as the circular pair shown in Figure 9.5.

If you steal property,
you must report its
fair market value in
your income in the
year you steal it
unless in the same
year, you return it to
its rightful owner.
Your Federal Income
Tax, IRS Publication
17, 2009.

Exercise 9.9. [?] Define an imperative-style procedure, mlist-reverse! , that re-
verses the elements of a list. Is it possible to implement a mlist-reverse! pro-
cedure that is asymptotically faster than the list-reverse procedure from Exam-
ple 5.4?

Exercise 9.10. [??] Define a procedure mlist-aliases? that takes as input two
mutable lists and outputs true if and only if there are any mcons cells shared
between the two lists.

9.4.2 Imperative Control Structures
The imperative style of programming makes progress by using assignments to
manipulate state. In many cases, solving a problem requires repeated opera-
tions. With functional programming, this is done using recursive definitions.
We make progress towards a base case by passing in different values for the
operands with each recursive application. With imperative programming, we
can make progress by changing state repeatedly without needing to pass in dif-
ferent operands.

A common control structure in imperative programming is a while loop. A while while loop

loop has a test condition and a body. The test condition is a predicate. If it
evaluates to true, the while loop body is executed. Then, the test condition is
evaluated again. The while loop continues to execute until the test condition

evaluates to false.

We can define while as a procedure that takes as input two procedures, a test
procedure and a body procedure, each of which take no parameters. Even though
the test and body procedures take no parameters, they need to be procedures in-
stead of expressions, since every iteration of the loop should re-evaluate the test
and body expressions of the passed procedures.

(define (while test body)
(if (test)

(begin (body) (while test body))
(void))) ; no result value

We can use the while procedure to implement Fibonacci similarly to the fast-
fibo procedure:

(define (fibo-while n)
(let ((a 1) (b 1))

(while (lambda () (> n 2))
(lambda () (set! b (+ a b))

(set! a (− b a))
(set! n (− n 1))))

b))

The final value of b is the result of the fibo-while procedure. In each iteration, the
body procedure is applied, updating the values of a and b to the next Fibonacci
numbers.

The value assigned to a is computed as (− b a) instead of b. The reason for this
is the previous assignment expression has already changed the value of b, by
adding a to it. Since the next value of a should be the old value of b, we can
find the necessary value by subtracting a. The fact that the value of a variable
can change depending on when it is used often makes imperative programming
trickier than functional programming.

An alternative approach, which would save the need to do subtraction, is to store
the old value in a temporary value:

(lambda ()
(let ((oldb b))

(set! b (+ a b))
(set! a oldb)
(set! n (− n 1))))

Many programming languages designed to support imperative programming
provide control constructs similar to the while procedure defined above. For ex-
ample, here is a version of the procedure in the Python programming language:

def fibonacci (n):
a = 1
b = 1
while n > 2:

a, b = b, a + b
n = n− 1

return b

We will use Python starting in Chapter 11, although you can probably guess what
most of this procedure means without knowing Python.

The most interesting statement is the double assignment: a, b = b, a + b. This
assigns the new value of a to the old value of b, and the new value of b to the sum
of the old values of a and b. Without the double assignment operator, it would
be necessary to store the old value of b in a new variable so it can be assigned to
a after updating b to the new value.

Exercise 9.11. Define the mlist-map! example from the previous section using
while.

Exercise 9.12. Another common imperative programming structure is a repeat-
until loop. Define a repeat-until procedure that takes two inputs, a body proce- repeat-until

dure and a test procedure. The procedure should evaluate the body procedure
repeatedly, until the test procedure evaluates to a true value. For example, using
repeat-until we could define factorial as:

(define (factorial n)
(let ((fact 1))

(repeat-until
(lambda () (set! fact (∗ fact n)) (set! n (− n 1)))
(lambda () (< n 1)))

fact))

Exercise 9.13. [??] Improve the efficiency of the indexing procedures from Sec-
tion 8.2.3 by using mutation. Start by defining a mutable binary tree abstraction,
and then use this and the MutableList data type to implement an imperative-
style insert-into-index! procedure that mutates the input index by adding a new
word-position pair to it. Then, define an efficient merge-index! procedure that
takes two mutable indexes as its inputs and modifies the first index to incor-
porate all word occurrences in the second index. Analyze the impact of your
changes on the running time of indexing a collection of documents.

9.5 Summary
Adding the ability to change the value associated with a name complicates our
evaluation rules, but enables simpler and more efficient solutions to many prob-
lems. Mutation allows us to efficiently manipulate larger data structures since it
is not necessary to copy the data structure to make changes to it.

Once we add assignment to our language, the order in which things happen
affects the value of some expressions. Instead of evaluating expressions using
substitution, we now need to always evaluate an expression in a particular exe-
cution environment.

The problem with mutation is that it makes it much tougher to reason about
the meaning of an expression. In the next chapter, we introduce a new kind of
abstraction that packages procedures with the state they manipulate. This helps
manage some of the complexity resulting from mutation by limiting the places
where data may be accessed and modified.

